

## BNF DA LINGUAGEM DE PROGRAMAÇÃO LINGPON 1.2

```
1 <program> ::= <fbes> <inst> <strategy> <rules> <main>
2           | <fbes> <inst> <strategy> <rules>
3
4 <fbes> ::= <fbe>
5         | <fbe> <fbes>
6
7 <fbe> ::= FBE <id> <fbe_body> END_FBE
8         | FBE <id> <extends> <id> <fbe_body> END_FBE
9
10 <fbe_body> ::= <decl_attributes> <decl_methods> <fbeRules>
11            | <decl_attributes> <decl_methods>
12            | <decl_attributes>
13
14 <inst> ::= INST <declarations> END_INST
15
16 <strategy> ::= STRATEGY <strategy_declaration> END_STRATEGY
17
18 <strategy_declaration> ::= NO_ONE
19                        | BREADTH
20                        | DEPTH
21
22 <declarations> ::= <declaration>
23                | <declaration> <declarations>
24
25 <declaration> ::= <type> <ids>
26
27 <ids> ::= <id>
28        | <id> COMMA <ids>
29
30 <rules> ::= <rule>
31         | <rule> <rules>
32         | <formRule>
33         | <formRule> <rules>
34
35 <rule> ::= RULE <rule_body> END_RULE
36         | RULE <id> <rule_body> END_RULE
37         | RULE <depends> <id> <rule_body> END_RULE
38         | RULE <id> <depends> <id> <rule_body> END_RULE
39
40 <formRule> ::= FORM_RULE <rule_body> END_FORM_RULE
41            | FORM_RULE <id> <rule_body> END_FORM_RULE
42            | FORM_RULE <depends> <id> <rule_body>
43              END_FORM_RULE
44            | FORM_RULE <id> <depends> <id> <rule_body>
45              END_FORM_RULE
46
47 <fbeRules> ::= <fbeRule>
48            | <fbeRule> <fbeRules>
49
50 <fbeRule> ::= FBE_RULE <rule_body> END_FBE_RULE
51           | FBE_RULE <id> <rule_body> END_FBE_RULE
52           | FBE_RULE <depends> <id> <rule_body> END_FBE_RULE
53           | FBE_RULE <id> <depends> <id> <rule_body>
54 END_FBE_RULE
55
56 <depends> ::= DEPENDS
57
58 <extends> ::= EXTENDS
```

---

```

59
60 <rule_body> ::= <decl_condition> <decl_action>
61 | <decl_properties> <decl_condition> <decl_action>
62
63 <decl_properties> ::= PROPERTIES <properties_body> END_PROPERTIES
64
65 <properties_body> ::= <properties_type> <value>
66
67 <properties_type> ::= PRIORITY
68 | KEEPER
69
70 <decl_condition> ::= CONDITION <condition_body> END_CONDITION
71 | CONDITION <id> <condition_body> END_CONDITION
72
73 <condition_body> ::= <subcondition> <operator> <condition_body>
74 | <subcondition>
75
76 <operator> ::= AND
77 | OR
78
79 <subcondition> ::= SUBCONDITION <id> <subcondition_body>
80 END_SUBCONDITION
81
82 <subcondition_body> ::= <premise> AND <subcondition_body>
83 | <premise>
84
85 <premise> ::= PREMISE <exp>
86 | PREMISE <id> <exp>
87 | PREMISE IMP <exp>
88 | PREMISE IMP <id> <exp>
89
90 <exp> ::= <fator> <comp> <fator>
91
92 <comp> ::= EQ
93 | NE
94 | LT
95 | GT
96 | LE
97 | GE
98
99 <fator> ::= <id>
100 | NUMBER
101 | boolean
102 | FLOATVALUE
103 | STRINGVALUE
104 | CHARVALUE
105
106 <boolean> ::= TRUE
107 | FALSE
108
109 <decl_action> ::= ACTION <action_body> END_ACTION
110 | ACTION <id> <action_body> END_ACTION
111
112 <action_body> ::= <action_elements> <action_body>
113 | <action_elements>
114
115 <action_elements> ::= <instigation>
116 | <method_use>
117 | <exp> SEMICOLON
118
119 <instigation> ::= INSTIGATION <method_use>

```

---

```

120         | INSTIGATION <id> <method_use>
121
122 <method_use> ::= <id> LP RP SEMICOLON
123
124 <id> ::= ID
125         | ID POINT <id>
126
127 <decl_attributes> ::= ATTRIBUTES <attributes> END_ATTRIBUTES
128
129 <attributes> ::= <attributes_body>
130         | <attributes_body> <attributes>
131
132 <attributes_body> ::= <type> <id> <value>
133         | <type> <id> SEMICOLON
134
135 <type> ::= BOOLEAN
136         | INTEGER
137         | PFLOAT
138         | STRING
139         | CHAR
140         | <id>
141
142 <value> ::= NUMBER
143         | <boolean>
144         | <id>
145         | FLOATVALUE
146         | STRINGVALUE
147         | CHARVALUE
148
149 <decl_methods> ::= METHODS <methods> END_METHODS
150
151 <methods> ::= <method_body>
152         | <method_body> <methods>
153
154 <method_body> ::= METHOD <id> LP <id> ASSIGN <id>
155         <method_operator> <value> RP
156         | METHOD <id> LP <id> ASSIGN <id>
157         <method_operator> <id> RP
158         | METHOD <id> LP <id> ASSIGN <value> RP
159         | METHOD <id> LP <id> ASSIGN <id> RP
160         | METHOD <id> LP <id> RP INNER_CODE_METHOD
161         | METHOD <id> LP RP INNER_CODE_METHOD
162
163 <method_operator> ::= PLUS
164         | MINUS
165         | MULT
166         | DIV
167
168 <main> ::= MAIN INNER_CODE_MAIN

```

**Código 1: BNF da linguagem de programação LingPON 1.2**  
**Fonte: Adaptado de Santos, 2017**